Лекция 18. Рекурсивный спуск


Метод рекурсивного спуска позволяет анализировать цепочки символов и проверять, являются ли они цепочками языка, описанного с помощью БНФ или нет. У метода есть ограничения и не для всяких метаформул есть возможность проверить цепочку рекурсивным спуском. Точное определение условий, когда метод применим, выходит за рамки нашего курса. Мы рассмотрим его в действии при решении задач, не решая сложных задач, где надо использовать другие методы.


Пусть даны БНФ:

<формула>::= <терм> | (<формула><знак><формула>)
 <знак>::= + | – | *

<терм>::= <имя> | <число>
<имя>::= <буква> | <имя><буква> | <имя><цифра>

<число>::= <цифра> | <число><цифра>
<цифра>::= 0 | 1 | … | 9

<буква> ::= a | b | … | z

Проверим с помощью рекурсивного спуска, является ли входной текст формулой. Идея состоит с том, что для каждой метапеременной создается своя собственная логическая функция, задача которой считать максимально длинную подцепочку, удовлетворяющую определения метасимвола. Если это удается, функция возвращает true, иначе false. При анализе стараются запоминать место, где возникла ошибка (несоответствие текста и БНФ) и выдавать диагностику, т. е. сообщать, в чем именно заключается несоответствие.

Program RecSpusk(input, output);

Const
eoi='.'; {признак конца входного текста}

Var 
ch:char; {последний прочитанный символ}


poz:integer; {номер текущей позиции}

        b:boolean;

{опережающие описания}

Function ReadFormula:Boolean; forward;

Function ReadZnak:Boolean; forward;

Function ReadTerm:Boolean; forward;

Function ReadImya:Boolean; forward;

Function ReadChislo:Boolean; forward;

Function ReadFormula{:Boolean};

Var b:boolean;

Begin
{перед вызовом этой функции в глобальной переменной ch находится 1ый символ цепочки, которую надо проверить, является ли она формулой}


If  (ch in ['0'.. '9']) or (ch in ['a'.. 'z']) then begin


{первая альтернатива: формула это терм}



ReadFormula:= ReadTerm

End else begin



If ch = '(' then begin




{вторая альтернатива: в формуле скобки}




read(ch); poz:=poz+1;




b:= ReadFormula;




if b then b:=ReadZnak;




if b then b:= ReadFormula;




if b then b:= ch = ')';




if b then begin





read(ch); poz:=poz+1;




end else begin




Writeln('Неожиданный символ в позиции ', poz, '. Ожидалась закрывающая скобка')

end;




ReadFormula:=b;



End else begin




If ch=eoi then begin





ReadFormula:=false;





Writeln('Неожиданный конец ввода в позиции ', poz)




End Else begin





ReadFormula:=false;





Writeln('Недопустимый символ в позиции ', poz)




End;



End


End;

End;

Function ReadZnak{:Boolean};

Begin


If ch in ['+', '-', '*'] then begin



ReadZnak:= true; read(ch); poz:=poz+1;


End Else begin

      Writeln('В позиции ', poz, ' ожидался знак, но вместо него встретился символ:', ch);

              readZnak:=false;


End

End;

Function ReadTerm:boolean;

Begin


If (ch>='0') and (ch <='9') then ReadTerm:= ReadChislo

Else



If (ch>='a') and (ch <='z') then ReadTerm:=ReadImya


Else begin
Writeln('В позиции ', poz, ' ожидался терм, но он не может начинаться с символа:', ch);

ReadTerm:=false



End

End;

Function ReadImya{:Boolean};

Begin


If (ch>='a') and (ch <='z') then begin



Repeat




Read(ch); poz:=poz+1;



Until not ((ch in ['a'..'z']) or (ch in ['0'..'9']));



ReadImya:=true


End else begin



ReadImya:=false;



Writeln('В позиции ', poz, ' ожидалось имя, но оно не может начинаться с символа:', ch);


end

End;

Function ReadChislo{:Boolean};

Begin


If (ch>='0') and (ch <='9') then begin



Repeat




Read(ch); poz:=poz+1;



Until  not (ch in ['0'..'9']);



ReadChislo:=true


End else begin



ReadChislo:=false;



Writeln('В позиции ', poz, ' ожидалось число, но оно не может начинаться с символа:', ch);


end

End;

Begin


Poz:=1;


Read(ch);


b:=ReadFormula();

if b and (ch<>eoi) then

            
Writeln('Неожиданный символ в позиции ', poz, '. Ожидался конец ввода.')

else
            
writeln(b);

End.

Рекурсивный спуск работает, если можно по текущему символу определить, по какой из альтернатив следует продолжать анализ. Пример, ситуации, когда это не выполняется:

<цепочка>::= <нули> | <нули> <нуль с единицей>
<нули> ::= 0{0}

<нуль с единицей> ::= 01

Здесь невозможно определить, какое из двух правил применять, если текущий символ 0. Либо это продолжаются <нули>, либо это начался <нуль с единицей>. Рекурсивный спуск неправильно проанализирует цепочки, оканчивающиеся на 1. Заметим, если описать тот же самый язык другими метаформулами, то рекурсивный спуск будет работать:
<цепочка>::= <нули> [1]

<нули> ::= 00{0}

